
Calculating a Backtracking Algorithm:

An Exercise in Monadic Program Derivation

SHIN-CHENG MU, Academia Sinica, Taiwan

Equational reasoning is among the most important tools that functional programming provides us. Curiously,

relatively less attention has been paid to reasoning about monadic programs. In this report we derive a back-

tracking algorithm for problem specifications that use a monadic unfold to generate possible solutions, which

are filtered using a scanl-like predicate. We develop theorems that convert a variation of scanl to a foldr that

uses the state monad, as well as theorems constructing hylomorphism. The algorithm is used to solve the

n-queens puzzle, our running example. The aim is to develop theorems and patterns useful for the derivation

of monadic programs, focusing on the intricate interaction between state and non-determinism.

1 INTRODUCTION

Equational reasoning is among the many gifts that functional programming offers us. Functional
programs preserve a rich set of mathematical properties, which not only helps to prove properties
about programs in a relatively simple and elegant manner, but also aids the development of pro-
grams. One may refine a clear but inefficient specification, stepwise through equational reasoning,
to an efficient program whose correctness may not be obvious without such a derivation.
It is misleading if one says that functional programming does not allow side effects. In fact, even

a purely functional language may allow a variety of side effects — in a rigorous, mathematically
manageable manner. Since the introduction of monads into the functional programming commu-
nity [Moggi 1989; Wadler 1992], it has become the main framework in which effects are modelled.
Various monads were developed for different effects, from general ones such as IO, state, non-
determinism, exception, continuation, environment passing, to specific purposes such as parsing.
Numerous research were also devoted to producing practical monadic programs.
It is also a wrong impression that impure programs are bound to be difficult to reason about.

In fact, the laws of monads and their operators are sufficient to prove quite a number of useful
properties about monadic programs. The validity of these properties, proved using only these
laws, is independent from the particular implementation of the monad.
This report follows the trail of Hutton and Fulger [2008] and Gibbons and Hinze [2011], aiming

to develop theorems and patterns that are useful for reasoning about monadic programs. We focus
on two effects — non-determinism and state. In this report we consider problem specifications that
use a monadic unfold to generate possible solutions, which are filtered using a scanl-like predicate.
We develop theorems that convert a variation of scanl to a foldr that uses the state monad, as well
as theorems constructing hylomorphism. The algorithm is used to solve the n-queens puzzle, our
running example.
While the interaction between non-determinism and state is known to be intricate, when each

non-deterministic branch has its own local state, we get a relatively well-behaved monad that
provides a rich collection of properties to work with. The situation when the state is global and
shared by all non-deterministic branches is much more complex, and is dealt with in a subsequent
paper [Pauwels et al. 2019].

2 Shin-Cheng Mu

2 MONAD AND EFFECT OPERATORS

A monad consists of a type constructorm :: ∗ → ∗ and two operators return and “bind” (>>=), often
modelled by the following Haskell type class declaration:1

class Monad m where

return :: a → m a

(>>=) ::m a → (a → m b) → m b .

They are supposed to satisfy the following monad laws:

return x >>= f = f x , (1)

m >>= return = m , (2)

(m >>= f ) >>= g = m >>= (_x → f x >>= g) . (3)

We also define m1 >> m2 = m1 >>= const m2, which has type (>>) :: m a → m b → m b. Kleisli
composition, denoted by (>=>), composes two monadic operations a → m b and b → m c into an
operation a → m c. The operator ( 〈$〉) applies a pure function to a monad.

(>=>) ::Monad m ⇒ (a → m b) → (b → m c) → a → m c

(f >=> g) x = f x >>= g ,

( 〈$〉) ::Monad m ⇒ (a → b) → m a → m b

f 〈$〉 n = n >>= (return · f ) .

The following properties can be proved from their definitions and the monad laws:

(f · g) 〈$〉 m = f 〈$〉 (g 〈$〉 m) , (4)

(f 〈$〉 m) >>= g = m >>= (g · f ) , (5)

f 〈$〉 (m >>= k) = m >>= (_x → f 〈$〉 k x) , x not free in f . (6)

Effect and Effect Operators. Monads are used to model effects, and each effect comes with its
collection of operators. For example, to model non-determinism we assume two operators ∅ and
(8), respectively modeling failure and choice. A state effect comes with operators get and put,
which respectively reads from and writes to an unnamed state variable.

A programmay involve more than one effect. In Haskell, the type class constraintMonadPlus in
the type of a program denotes that the programmay use ∅ or (8), and possibly other effects, while
MonadState s denotes that it may use get and put. Some theorems in this report, however, apply
only to programs that, for example, use non-determinism and no other effects. In such cases we will
note in text that the theorem applies only to programs “whose only effect is non-determinism.” The
set of effects a program uses can always be statically inferred by syntax.

Total, Finite Programs. Like in other literature on program derivation, we assume a set-theoretic
semantics in which functions are total. We thus have the following laws regarding branching:

f (if p then e1 else e2) = if p then f e1 else f e2 , (7)

if p then (_x → e1) else (_x → e2) = _x → if p then e1 else e2 . (8)

1This report uses type classes to be explicit about the effects a program uses. For example, programs using non-determinism

are labelled with constraint MonadPlus m. The style of reasoning proposed in this report is not tied to type classes or

Haskell, and we do not strictly follow the particularities of type classes in the current Haskell standard. For example,

we overlook the particularities that a Monad must also be Applicative, MonadPlus be Alternative, and that functional

dependency is needed in a number of places in this report.

Calculating a Backtracking Algorithm 3

Lists in this report are inductive types, and unfolds generate finite lists too. Non-deterministic
choices are finitely branching. Given a concrete input, a function always expands to a finitely-
sized expression consisting of syntax allowed by its type. We may therefore prove properties of a
monadic program by structural induction over its syntax.

3 EXAMPLE: THE n-QUEENS PROBLEM

Reasoning about monadic programs gets more interesting when more than one effect is involved.
Backtracking algorithmsmake good examples of programs that are stateful and non-deterministic,
and the n-queens problem, also dealt with by Gibbons and Hinze [2011], is among the most well-
known examples of backtracking.2

In this section we present a specification of the problem, before transforming it into the form
unfoldM p f >=> filt (all ok · scanl+ (⊕) st) (whose components will be defined later), which is the
general form of problems we will deal with in this report.

3.1 Non-Determinism

Since the n-queens problem will be specified by a non-deterministic program, we discuss non-
determinism before presenting the specification. We assume two operators ∅ and (8):

class Monad m ⇒ MonadPlus m where

∅ ::m a

(8) ::m a → m a → m a .

The former denotes failure, while m 8 n denotes that the computation may yield either m or n.
What laws they should satisfy, however, can be a tricky issue. As discussed by Kiselyov [2015], it
eventually comes down to what we use the monad for. It is usually expected that (8) and ∅ form
a monoid. That is, (8) is associative, with ∅ as its zero:

(m 8 n) 8 k = m 8 (n 8 k) , (9)

∅ 8 m = m = m 8 ∅ . (10)

It is also assumed that monadic bind distributes into (8) from the end, while ∅ is a left zero for
(>>=):

left-distributivity : (m1 8 m2) >>= f = (m1 >>= f ) 8 (m2 >>= f ) , (11)

left-zero : ∅ >>= f = ∅ . (12)

Wewill refer to the laws (9), (10), (11), (12) collectively as the nondeterminism laws. Other properties
regarding ∅ and (8) will be introduced when needed.
The monadic function filt p x returns x if p x holds, and fails otherwise:

filt ::MonadPlus m ⇒ (a → Bool) → a → m a

filt p x = guard (p x) >> return x ,

where guard is a standard monadic function defined by:

guard ::MonadPlus m ⇒ Bool → m ()

guard b = if b then return () else ∅ .

The following properties allow us to move guard around. Their proofs are given in Appendix A.

guard (p ∧ q) = guard p >> guard q , (13)

2Curiously, Gibbons and Hinze [2011] did not finish their derivation and stopped at a program that exhaustively generates

all permutations and tests each of them. Perhaps it was sufficient to demonstrate their point.

0 1 2 3 4 5 6 7
0 . . . . . & . .

1 . . . & . . . .

2 . . . . . . & .

3 & . . . . . . .

4 . . . . . . . &

5 . & . . . . . .

6 . . . . & . . .

7 . . & . . . . .

(a)

0 1 2 3 4 5 6 7
0 0 1 2 3 4 . . .

1 1 2 3 4 . . . .

2 2 3 4 . . . . .

3 3 4 . . . . . .

4 4 . . . . . . .

5 . . . . . . . 12
6 . . . . . . 12 13
7 . . . . . 12 13 14

(b)

0 1 2 3 4 5 6 7
0 0 −1 . . . −5 −6 −7
1 . 0 −1 . . . −5 −6
2 . . 0 −1 . . . −5
3 3 . . 0 . . . .

4 4 3 . . 0 . . .

5 5 4 3 . . 0 . .

6 6 5 4 3 . . 0 .

7 7 6 5 4 3 . . 0

(c)

Fig. 1. (a) This placement can be represented by [3, 5, 7, 1, 6, 0, 2, 4]. (b) Up diagonals. (c) Down diagonals.

guard p >> (f 〈$〉 m) = f 〈$〉 (guard p >> m) . (14)

guard p >> m = m >>= (_x → guard p >> return x), if m >> ∅ = ∅ . (15)

3.2 Specification

The aim of the puzzle is to place n queens on a n by n chess board such that no two queens can
attack each other. Given n, we number the rows and columns by [0 . . n − 1]. Since all queens
should be placed on distinct rows and distinct columns, a potential solution can be represented by
a permutation xs of the list [0 . . n−1], such that xs !! i = j denotes that the queen on the 8th column
is placed on the 9 th row (see Figure 1(a)). In this representation queens cannot be put on the same
row or column, and the problem is reduced to filtering, among permutations of [0 . . n − 1], those
placements in which no two queens are put on the same diagonal. The specification can be written
as a non-deterministic program:

queens ::MonadPlus m ⇒ Int → m [ Int]

queens n = perm [0 . . n − 1] >>= filt safe ,

where perm non-deterministically computes a permutation of its input, and the pure function
safe :: [ Int] → Bool determines whether no queens are on the same diagonal.
This specification of queens generates all the permutations, before checking them one by one,

in two separate phases. We wish to fuse the two phases and produce a faster implementation. The
overall idea is to define perm in terms of an unfold, transform filt safe into a fold, and fuse the two
phases into a hylomorphism [Meijer et al. 1991]. During the fusion, some non-safe choices can be
pruned off earlier, speeding up the computation.

Permutation. The monadic function perm can be written both as a fold or an unfold. For this
problem we choose the latter. The function select non-deterministically splits a list into a pair
containing one chosen element and the rest:

select ::MonadPlus m ⇒ [a] → m (a, [a]) .

select [ ] = ∅

select (x : xs) = return (x, xs) 8 ((id × (x:)) 〈$〉 select xs) ,

where (f × g) (x, y) = (f x, g y). For example, select [1, 2, 3] yields one of (1, [2, 3]), (2, [1, 3])
and (3, [1, 2]). The function call unfoldM p f y generates a list [a] from a seed y :: b. If p y holds,
the generation stops. Otherwise an element and a new seed is generated using f . It is like the usual
unfoldr apart from that f , and thus the result, is monadic:

unfoldM ::Monad m ⇒ (b → Bool) → (b → m (a, b)) → b → m [a]

unfoldM p f y | p y = return [ ]

| otherwise = f y >>= _(x, z) → (x:) 〈$〉 unfoldM p f z .

Given these definitions, perm can be defined by:

perm ::MonadPlus m ⇒ [a] → m [a]

perm = unfoldM null select .

3.3 Safety Check in a scanl

We have yet to define safe. Representing a placement as a permutation allows an easy way to check
whether two queens are put on the same diagonal. An 8 by 8 chess board has 15 up diagonals (those
running between bottom-left and top-right). Let them be indexed by [0 . . 14] (see Figure 1(b)). If
we apply zipWith (+) [0 . . ] to a permutation, we get the indices of the up-diagonals where the
chess pieces are placed. Similarly, there are 15 down diagonals (those running between top-left and
bottom right). By applying zipWith (−) [0 . . ] to a permutation, we get the indices of their down-
diagonals (indexed by [−7 . . 7]. See Figure 1(c)). A placement is safe if the diagonals contain no
duplicates:

ups, downs :: [ Int] → [ Int]

ups xs = zipWith (+) [0 . . ] xs ,

downs xs = zipWith (−) [0 . . ] xs ,

safe :: [ Int] → Bool

safe xs = nodup (ups xs) ∧ nodup (downs xs) ,

where nodup :: Eq a ⇒ [a] → Bool determines whether there is no duplication in a list.
The eventual goal is to transform filt safe into a foldr, to be fused with perm, an unfold that

generates a list from left to right. In order to do so, it helps if safe can be expressed in a computation
that processes the list left-to-right, that is, a foldl or a scanl. To derive such a definition we use the
standard trick — introducing accumulating parameters, and generalising safe to safeAcc below:

safeAcc :: (Int, [ Int], [ Int]) → [ Int] → Bool

safeAcc (i, us, ds) xs = nodup us′ ∧ nodup ds′ ∧

all (∉ us) us′ ∧ all (∉ ds) ds′ ,

where us′ = zipWith (+) [ i . . ] xs

ds′ = zipWith (−) [ i . . ] xs .

It is a generalisation because safe = safeAcc (0, [ ], [ ]). By plain functional calculation, one may
conclude that safeAcc can be defined using a variation of scanl:

safeAcc (i, us, ds) = all ok · scanl+ (⊕) (i, us, ds) ,

where (i, us, ds) ⊕ x = (i + 1, (i + x : us), (i − x : ds))

ok (i, (x : us), (y : ds)) = x ∉ us ∧ y ∉ ds ,

where all p = foldr (∧) True · map p and scanl+ is like the standard scanl, but applies foldl to all
non-empty prefixes of a list. It can be specified by:

scanl+ :: (b → a → b) → b → [a] → [b ]

scanl+ (⊕) st = tail · scanl (⊕) st ,

and it also adopts an inductive definition:

Technical Report TR-IIS-19-003, Institute of Information Science, Academia Sinica. Publication date: June 2019.



6 Shin-Cheng Mu

scanl+ (⊕) st [ ] = [ ]

scanl+ (⊕) st (x : xs) = (st ⊕ x) : scanl+ (⊕) (st ⊕ x) xs .

Operationally, safeAcc examines the list from left to right, while keeping a state (i, us, ds), where
i is the current position being examined, while us and ds are respectively indices of all the up and
down diagonals encountered so far. Indeed, in a function call scanl+ (⊕) st , the value st can be
seen as a “state” that is explicitly carried around. This naturally leads to the idea: can we convert
a scanl+ to a monadic program that stores st in its state? This is the goal of the next section.
As a summary of this section, after defining queens, we have transformed it into the following

form:

unfoldM p f >=> filt (all ok · scanl+ (⊕) st) .

This is the form of problems we will consider for the rest of this report: problems whose solutions
are generated by an monadic unfold, before being filtered by an filt that takes the result of a scanl+.

4 FROM PURE TO STATEFUL scanl

The aim of this section is to turn the filtering phase filt (all ok · scanl+ (⊕) st) into a foldr . For
that we introduce a state monad to pass the state around.
The state effect provides two operators:

class Monad m ⇒ MonadState s m where

get ::m s

put :: s → m () ,

where get retrieves the state, while put overwrites the state by the given value. They are supposed
to satisfy the state laws:

put-put : put st >> put st ′ = put st ′ , (16)

put-get : put st >> get = put st >> return st , (17)

get-put : get >>= put = return () , (18)

get-get : get >>= (_st → get >>= k st) = get >>= (_st → k st st) . (19)

4.1 From scanl+ to monadic foldr

Consider the following monadic variation of scanl:

scanlM ::MonadState s m ⇒ (s → a → s) → s → [a] → m [ s ]

scanlM (⊕) st xs = put st >> foldr (⊗) (return [ ]) xs

where x ⊗ n = get >>= _st → let st ′ = st ⊕ x

in (st ′:) 〈$〉 (put st ′ >> n) .

It behaves like scanl+, but stores the accumulated information in amonadic state, which is retrieved
and stored in each step. The main body of the computation is implemented using a foldr .
To relate scanl+ and scanlM , one would like to have return (scanl+ (⊕) st xs) = scanlM (⊕) st xs.

However, the lefthand side does not alter the state, while the righthand side does. One of the ways
to make the equality hold is to manually backup and restore the state. Define

protect ::MonadState s m ⇒ m b → m b

protect n = get >>= _ini → n >>= _x → put ini >> return x ,

We have

Technical Report TR-IIS-19-003, Institute of Information Science, Academia Sinica. Publication date: June 2019.

Theorem 4.1. For all (⊕) :: (s → a → s), st :: s, and xs :: [a],

return (scanl+ (⊕) st xs) = protect (scanlM (⊕) st xs) .

Proof. By induction on xs. We present the case xs := x : xs.

protect (scanlM (⊕) st (x : xs))

= { expanding definitions, let st ′ = st ⊕ x }

get >>= _ini → put st >> get >>= _st →

((st ′:) 〈$〉 (put st ′ >> foldr (⊗) (return [ ]) xs)) >>= _r →

put ini >> return r

= { by put-get (17) }

get >>= _ini → put st >>

((st ′:) 〈$〉 (put st ′ >> foldr (⊗) (return [ ]) xs)) >>= _r →

put ini >> return r

= { by (6) }

(st ′:) 〈$〉 (get >>= _ini → put st >> put st ′ >>

foldr (⊗) (return [ ]) xs >>= _r →

put ini >> return r)

= { by put-put (16) }

(st ′:) 〈$〉 (get >>= _ini → put st ′ >> foldr (⊗) (return [ ]) xs

>>= _r → put ini >> return r)

= { definitions of scanlM and protect }

(st ′:) 〈$〉 protect (scanlM (⊕) st ′ xs)

= { induction }

(st ′:) 〈$〉 return (scanl+ (⊕) st ′ xs)

= return ((st ⊕ x) : scanl+ (⊕) (st ⊕ x) xs)

= return (scanl+ (⊕) st (x : xs)) .

�

This proof is instructive due to the use of properties (16) and (17), and that (st ′:), being a pure
function, can be easily moved around using (6).
We have learned that scanl+ (⊕) st can be turned into scanlM (⊕) st, defined in terms of a

stateful foldr. In the definition, state is the only effect involved. The next task is to transform
filt (scanl+ (⊕) st) into a foldr. The operator filt is defined using non-determinism. Hence the
transformation involves the interaction between two effects.

4.2 Right-Distributivity and Local State

We now digress a little to discuss one form of interaction between non-determinism and state. In
this report, we wish that the following two additional properties are valid:

right-distributivity : m >>= (_x → f1 x 8 f2 x) = (m >>= f1) 8 (m >>= f2) , (20)

right-zero : m >> ∅ = ∅ . (21)

Note that the two properties hold for some monads with non-determinism, but not all. With some
implementations of the monad, it is likely that in the lefthand side of (20), the effect ofm happens
once, while in the righthand side it happens twice. In (21), the m on the lefthand side may incur
some effects that do not happen in the righthand side.

Having (20) and (21) leads to profound consequences on the semantics and implementation of
monadic programs. To beginwith, (20) implies that (8) be commutative. To see that, letm = m18m2

and f1 = f2 = return in (20). Implementation of such non-deterministic monads have been studied
by Kiselyov [2013].
When mixed with state, one consequence of (20) is that get >>= (_s → f1 s 8 f2 s) = (get >>=

f1 8 get >>= f2). That is, f1 and f2 get the same state regardless of whether get is performed outside
or inside the non-determinism branch. Similarly, (21) implies put s >> ∅ = ∅ — when a program
fails, the changes it performed on the state can be discarded. These requirements imply that each
non-determinism branch has its own copy of the state. Therefore, we will refer to (20) and (21) as
local state laws in this report — even though they do not explicitly mention state operators at all!
One monad satisfying the local state laws is M a = s → [ (a, s) ], which is the same monad one

gets by StateT s (ListT Identity) in the Monad Transformer Library [Gill and Kmett 2014]. With
effect handling [Kiselyov and Ishii 2015; Wu et al. 2012], the monad meets the requirements if we
run the handler for state before that for list.
The advantage of having the local state laws is that we get many useful properties, which make

this stateful non-determinism monad preferred for program calculation and reasoning. Recall, for
example, that (21) is the antecedent of (15). The result can be stronger: non-determinism commutes
with all other effects if we have local state laws.

Definition 4.2. Let m and n be two monadic programs such that x does not occur free in m, and
y does not occur free in n. We say m and n commute if

m >>= _x → n >>= _y → f x y =

n >>= _y → m >>= _x → f x y .
(22)

We say that m commutes with effect X if m commutes with any n whose only effects are X , and
that effects n and X commute if anym and n commute as long as their only effects are respectively
n and X .

Theorem 4.3. If right-distributivity (20) and right-zero (21) hold in addition to the monad laws

stated before, non-determinism commutes with any effect n .

Proof. Let m be a monadic program whose only effect is non-determinism, and stmt be any
monadic program. The aim is to prove that m and stmt commute. Induction on the structure ofm.
Case m := return e:

stmt >>= _x → return e >>= _y → f x y

= { monad law (1) }

stmt >>= _x → f x e

= { monad law (1) }

return e >>= _y → stmt >>= _x → f x y .

Case m :=m1 8 m2:

stmt >>= _x → (m1 8 m2) >>= _y → f x y

= { by (11) }

stmt >>= _x → (m1 >>= f x) 8 (m2 >>= f x)

= { by (20) }

(stmt >>= _x → m1 >>= f x) 8 (stmt >>= _x → m2 >>= f x)

= { induction }

(m1 >>= _y → stmt >>= _x → f x y) 8 (m2 >>= _y → stmt >>= _x → f x y)

Technical Report TR-IIS-19-003, Institute of Information Science, Academia Sinica. Publication date: June 2019.



Calculating a Backtracking Algorithm 9

= { by (11) }

(m1 8 m2) >>= _y → stmt >>= _x → f x y .

Case m := ∅:

stmt >>= _x → ∅ >>= _y → f x y

= { by (12) }

stmt >>= _x → ∅

= { by (21) }

∅

= { by (12) }

∅ >>= _y → stmt >>= _x → f x y .

�

Note. We briefly justify proofs by induction on the syntax tree. Finite monadic programs can be
represented by the free monad constructed out of return and the effect operators, which can be rep-
resented by an inductively defined data structure, and interpreted by effect handlers [Kiselyov and Ishii
2015; Kiselyov et al. 2013]. When we say two programs m1 and m2 are equal, we mean that they
have the same denotation when interpreted by the effect handlers of the corresponding effects, for
example, hdNondet (hdState s m1) = hdNondet (hdState s m2), where hdNondet and hdState are
respectively handlers for nondeterminism and state. Such equality can be proved by induction on
some sub-expression in m1 orm2, which are treated like any inductively defined data structure. A
more complete treatment is a work in progress. (End of Note)

4.3 Filtering Using a Stateful, Non-Deterministic Fold

Having dealt with scanl+ (⊕) st in Section 4.1, in this section we aim to turn a filter of the form
filt (all ok · scanl+ (⊕) st) to a stateful and non-deterministic foldr .
We calculate, for all ok, (⊕), st , and xs:

filt (all ok · scanl+ (⊕) st) xs

= guard (all ok (scanl+ (⊕) st xs)) >> return xs

= return (scanl+ (⊕) st xs) >>= _ys →

guard (all ok ys) >> return xs

= { Theorem 4.1, definition of protect , monad law }

get >>= _ini → scanlM (⊕) st xs >>= _ys → put ini >>

guard (all ok ys) >> return xs

= { Theorem 4.3: non-determinism commutes with state }

get >>= _ini → scanlM (⊕) st xs >>= _ys →

guard (all ok ys) >> put ini >> return xs

= { definition of protect , monad laws }

protect (scanlM (⊕) st xs >>= (guard · all ok) >> return xs) .

Recall that scanlM (⊕) st xs = put st >> foldr (⊗) (return [ ]) xs. The following theorem fuses a
monadic foldr with a guard that uses its result.

Theorem 4.4. Assume that state and non-determinism commute. Let (⊗) be defined as that in

scanlM for any given (⊕) :: s → a → s. We have that for all ok :: s → Bool and xs :: [a]:

foldr (⊗) (return [ ]) xs >>= (guard · all ok) >> return xs =

foldr (⊙) (return [ ]) xs ,

where x ⊙ m = get >>= _st → guard (ok (st ⊕ x)) >>

put (st ⊕ x) >> ((x:) 〈$〉 m) .

Proof. Unfortunatelywe cannot use a foldr fusion, since xs occurs free in _ys → guard (all ok ys)>>

return xs. Instead we use a simple induction on xs. For the case xs := x : xs:

(x ⊗ foldr (⊗) (return [ ]) xs) >>= (guard · all ok) >> return (x : xs)

= { definition of (⊗) }

get >>= _st →

(((st ⊕ x):) 〈$〉 (put (st ⊕ x) >> foldr (⊗) (return [ ]) xs)) >>=

(guard · all ok) >> return (x : xs)

= { monad laws, (5), and (6) }

get >>= _st → put (st ⊕ x) >>

foldr (⊗) (return [ ]) xs >>= _ys →

guard (all ok (st ⊕ x : ys)) >> return (x : xs)

= { since guard (p ∧ q) = guard q >> guard p }

get >>= _st → put (st ⊕ x) >>

foldr (⊗) (return [ ]) xs >>= _ys →

guard (ok (st ⊕ x)) >> guard (all ok ys) >>

return (x : xs)

= { assumption: nondeterminism commutes with state }

get >>= _st → guard (ok (st ⊕ x)) >> put (st ⊕ x) >>

foldr (⊗) (return [ ]) xs >>= _ys →

guard (all ok ys) >> return (x : xs)

= { monad laws and definition of ( 〈$〉) }

get >>= _st → guard (ok (st ⊕ x)) >> put (st ⊕ x) >>

(x:) 〈$〉 (foldr (⊗) (return [ ]) xs >>= _ys → guard (all ok ys) >> return xs)

= { induction }

get >>= _st → guard (ok (st ⊕ x)) >> put (st ⊕ x) >>

(x:) 〈$〉 foldr (⊙) (return [ ]) xs

= { definition of (⊙) }

foldr (⊙) (return [ ]) (x : xs) .

�

This proof is instructive due to extensive use of commutativity.
In summary, we now have this corollary performing filt (all ok · scanl+ (⊕) st) using a non-

deterministic and stateful foldr:

Corollary 4.5. Let (⊙) be defined as in Theorem 4.4. If state and non-determinism commute, we

have:

filt (all ok · scanl+ (⊕) st) xs =

protect (put st >> foldr (⊙) (return [ ]) xs) .

5 MONADIC HYLOMORPHISM

To recap what we have done, we started with a specification of the form unfoldM p f z >>=

filt (all ok · scanl+ (⊕) st), where f ::MonadPlus m ⇒ b → m (a, b), and have shown that

unfoldM p f z >>= filt (all ok · scanl+ (⊕) st)

= { Corollary 4.5, with (⊙) defined as in Theorem 4.4 }

unfoldM p f z >>= _xs → protect (put st >> foldr (⊙) (return [ ]) xs)

= { Theorem 4.3: nondeterminism commutes with state }

protect (put st >> unfoldM p f z >>= foldr (⊙) (return [ ])) .

The final task is to fuse unfoldM p f with foldr (⊙) (return [ ]).

5.1 Monadic Hylo-Fusion

In a pure setting, it is known that, provided that the unfolding phase terminates, foldr (⊗) e ·

unfoldr p f is the unique solution of hylo in the equation below [Hinze et al. 2015]:

hylo y | p y = e

| otherwise = let f y = (x, z) in x ⊗ hylo z .

Hylomorphisms with monadic folds and unfolds are a bit tricky. Pardo [2001] discussed hylomor-
phism for regular base functors, where the unfolding phase is monadic while the folding phase is
pure. As for the case when both phases are monadic, he noted “the drawback ... is that they cannot
be always transformed into a single function that avoids the construction of the intermediate data
structure.”
For our purpose, we focus our attention on lists, and have a theorem fusing the monadic unfold-

ing and folding phases under a side condition. Given (⊗) :: b → m c → m c, e :: c, p :: a → Bool,
and f :: a → m (b, a) (whereMonad m), consider the expression:

unfoldM p f >=> foldr (⊗) (return e) :: Monad m ⇒ a → m c .

The following theorem says that this combination of folding and unfolding can be fused into one,
with some side conditions:

Theorem 5.1. Let m :: ∗ → ∗ be in type class Monad. For all (⊗) :: a → m c → m c, e :: m c,

p :: b → Bool, and f :: b → m (a, c), we have that unfoldM p f >=> foldr (⊗) e = hyloM (⊗) e p f ,

defined by:

hyloM (⊗) e p f y | p y = e

| otherwise = f y >>= _(x, z) → x ⊗ hyloM (⊗) e p f z ,

if the relation (¬ · p) ? ·snd · (=<<) · f is well-founded (see the note below) and, for all k, we have

n >>= ((x⊗) · k) = x ⊗ (n >>= k) , (23)

where n abbreviates unfoldM p f z.

The “well-foundedness” condition essentially says that f eventually terminates — details to be
explained after the proof of this theorem. Condition (23) may look quite restrictive. In most cases
the author have seen, however, we can actually prove that (23) holds for an entire class of n that in-
cludes unfoldM p f z. In the application of this report, for example, the only effect of unfoldM p f z

is non-determinism, and we will prove in Lemma 5.2 that (23) holds for all n whose only effect is
non-determinism, for the particular operator we use in the n-queens problem.
We prove Theorem 5.1 below.

Proof. We start with showing that unfoldM p f >=> foldr (⊗) e is a fixed-point of the recursive
equations of hyloM . When p y holds, it is immediate that

return [ ] >>= foldr (⊗) e = e .

When ¬ (p y), we reason:

unfoldM p f y >>= foldr (⊗) e

= { definition of unfoldM , ¬ (p y) }

(f y >>= (_(x, z) → (x:) 〈$〉 unfoldM p f z)) >>= foldr (⊗) e

= { monad law and foldr }

f y >>= (_(x, z) → unfoldM p f z >>= _xs → x ⊗ foldr (⊗) e xs)

= { since n >>= ((x⊗) · k) = x ⊗ (n >>= k) where n = unfoldM p f z }

f y >>= (_(x, z) → x ⊗ (unfoldM p f z >>= foldr (⊗) e)) .

Now that unfoldM p f z >>= foldr (⊗) e is a fixed-point, wemay conclude that it equals hyloM (⊗) e p f

if the latter has a unique fixed-point, which is guaranteed by the well-foundedness condition. See
the note below. �

Note. Let q be a predicate, q? is a relation defined by { (x, x) | q x }. The parameter y in unfoldM

is called the seed used to generate the list. The relation (¬ · p) ? ·snd · (=<<) · f maps one seed to the
next seed (where (=<<) is (>>=) written reversed). If it is well-founded, intuitively speaking, the seed
generation cannot go on forever and p will eventually hold. It is known that inductive types (those
can be folded) and coinductive types (those can be unfolded) do not coincide in SET. To allow a
fold to be composed after an unfold, typically one moves to a semantics based on complete partial
orders. However, it was shown [Doornbos and Backhouse 1995] that, in Rel, when the relation
generating seeds is well-founded, hylo-equations do have unique solutions. One may thus stay
within a set-theoretic semantics. Such an approach is recently explored again [Hinze et al. 2015].
(End of Note)

Theorem 5.1 does not rely on the local state laws (20) and (21), and does not put restriction on n .
To apply the theorem to our particular case, we have to show that its preconditions hold for our
particular (⊙) — for that we will need (21) and perhaps also (20). In the lemma below we slightly
generalise (⊙) in Theorem 4.4:

Lemma 5.2. Assume that (21) holds. Given p :: a → s → Bool, next :: a → s → s, and res :: a →

b → b, define (⊙) as below:

(⊙) :: (MonadPlus m,MonadState s m) ⇒ a → m b → m b

x ⊙ m = get >>= _st → guard (p x st) >>

put (next x st) >> (res x 〈$〉 m) .

We have n >>= ((x⊙) · k) = x ⊙ (n >>= k), if n commutes with state.

Proof. We reason:

n >>= ((x⊙) · k)

= n >>= _y → x ⊙ k y

= { definition of (⊙) }

n >>= _y → get >>= _st →

guard (p x st) >> put (next x st) >> (res x 〈$〉 k y)

= { n commutes with state }

get >>= _st → n >>= _y →

guard (p x st) >> put (next x st) >> (res x 〈$〉 (k y))

= { by (15), since (21) holds }

get >>= _st → guard (p x st) >>

n >>= _y → put (next x st) >> (res x 〈$〉 (k y))

= { n commutes with state }

get >>= _st → guard (p x st) >>

put (next x st) >> n >>= _y → (res x 〈$〉 (k y))

= { properties of ( 〈$〉) }

get >>= _st → guard (p x st) >>

put (next x st) >> (res x 〈$〉 n >>= k)

= { definition of (⊙) }

x ⊙ (n >>= k) .

�

5.2 Summary, and Solving n-�eens

To conclude our derivation, a problem formulated as unfoldM p f z >>= filt (all ok · scanl+ (⊕) st)

can be solved by a hylomorphism. Define:

solve :: (MonadState s m,MonadPlus m) ⇒

(b → Bool) → (b → m (a, b)) → (s → Bool) → (s → a → s) → s → b → m [a]

solve p f ok (⊕) st z = protect (put st >> hyloM (⊙) (return [ ]) p f z) ,

where x ⊙ m = get >>= _st → guard (ok (st ⊕ x)) >>

put (st ⊕ x) >> ((x:) 〈$〉 m) .

Corollary 5.3. Given p :: b → Bool, f :: (MonadPlus m,MonadState s m) ⇒ b → m (a, b),

z :: b, ok :: s → Bool, (⊕) :: s → a → s, st :: s, If the relation (¬ · p) ? ·snd · (=<<) · f is well-founded, the
local state laws hold in addition to the other laws, and unfoldM p f z commutes with state, we have

unfoldM p f z >>= filt (all ok · scanl+ (⊕) st) =

solve p f ok (⊕) st z .

n-Queens Solved. Recall that

queens n = perm [0 . . n − 1] >>= filt safe

= unfoldM null select [0 . . n − 1] >>= filt (all ok · scanl+ (⊕) (0, [ ], [ ])) ,

where the auxiliary functions select , ok, (⊕) are defined in Section 3. The function select cannot
be applied forever since the length of the given list decreases after each call, and perm, using only
non-determinism, commutes with state. Therefore, Corollary 5.3 applies, and we have queens n =

solve null select ok (⊕) (0, [ ], [ ]) [0 . . n − 1]. Expanding the definitions we get:

queens :: (MonadPlus m,MonadState (Int, [ Int], [ Int]) m) ⇒ Int → m [ Int]

queens n = protect (put (0, [ ], [ ]) >> queensBody [0 . . n − 1]) ,

queensBody :: (MonadPlus m,MonadState (Int, [ Int], [ Int]) m) ⇒ [ Int] → m [ Int]

queensBody [ ] = return [ ]

queensBody xs = select xs >>= _(x, ys) →

get >>= _st → guard (ok (st ⊕ x)) >>

put (st ⊕ x) >> ((x:) 〈$〉 queensBody ys) ,

where (i, us, ds) ⊕ x = (1 + i, (i + x) : us, (i − x) : ds)

ok ( , u : us, d : ds) = (u ∉ us) ∧ (d ∉ ds) .

This completes the derivation of our backtracking algorithm for the n-queens problem.

6 CONCLUSIONS AND RELATED WORK

This report is a case study of reasoning and derivation of monadic programs. To study the interac-
tion between non-determinism and state, we construct backtracking algorithms solving problems
that can be specified in the form unfoldM f p >=> assert (all ok · scanl+ (⊕) st). The derivation of
the backtracking algorithmworks by fusing the two phases into a monadic hylomorphism. It turns
out that in derivations of programs using non-determinism and state, commutativity plays an im-
portant role. We assume in this report that the local state laws (right-distributivity and right-zero)
hold. In this scenario we have nicer properties at hand, and commutativity holds more generally.
The local state laws imply that each non-deterministic branch has its own state. It is cheap to

implement when the state can be represented by linked data structures, such as a tuple containing
lists, as in the n-queens example. When the state contains blocked data, such as a large array,
duplicating the state for each non-deterministic branch can be costly. Hence there is practical
need for sharing one global state among non-deterministic branches. When a monad supports
shared global state and non-determinism, commutativity of the two effects holds in limited cases.
The behaviour of the monad is much less intuitive, and might be considered awkward sometimes.
In a subsequent paper [Pauwels et al. 2019], we attempt to find out what algebraic laws we can
expect and how to reason with programs when the state is global.
Affeldt et al. [2019] modelled a hierarchy of monadic effects in Coq. The formalisation was ap-

plied to verify a number of equational proofs of monadic programs, including some of the proofs
in an earlier version of this report. A number of errors was found and reported to the author.

following-up this report. Thanks also go to Reynald Affeldt, David Nowak and Takafumi Saikawa

A MISCELLANEOUS PROOFS

Proofs of (13) and (14).

Proof. Proof of (13) relies only on property of if and conjunction:

guard (p ∧ q)

= if p ∧ q then return () else ∅

= if p then (if q then return () else ∅) else ∅

= if p then guard q else ∅

= guard p >> guard q .

To prove (14), surprisingly, we need only distributivity and not (12):

guard p >> (f 〈$〉 m)

= { definition of guard }

(if p then return () else ∅) >> (f 〈$〉 m)

= { by (7) }

if p then return () >> (f 〈$〉 m) else ∅ >>> (f 〈$〉 m)

= { by (3) and (1) }

if p then f 〈$〉 (return () >> m) else f 〈$〉 (∅ >> m)

= { by (7) }

f 〈$〉 ((if p then return () else ∅) >> m)

= { definition of guard }

f 〈$〉 (guard p >> m) .

�

Proof of (15).

Proof. We reason:

m >>= _x → guard p >> return x

= { definition of guard }

m >>= _x → (if p then return () else ∅) >> return x

= { by (7), with f n = m >>= _x → n >> return x }

if p then m >>= _x → return () >> return x

else m >>= _x → ∅ >> return x

= { since return () >> n = n and (12) }

if p then m else m >> ∅

= { assumption: m >> ∅ = ∅ }

if p then m else ∅

= { since return () >> n = n and (12) }

if p then return () >> m

else ∅ >> m

= { by (7), with f = (>>m) }

(if p then return () else ∅) >> m

= { definition of guard }

guard p >> m .

�

